# Contributor: Manolis Tzanidakis
# Contributor: Mark Rosenstand <mark@archlinux.org>
# Contributor: Thayer Williams <thayer@archlinux.org>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=libdaemon
pkgver=0.14
pkgrel=5
pkgdesc='Lightweight C library that eases the writing of UNIX daemons'
url='http://0pointer.de/lennart/projects/libdaemon/'
license=('LGPL')
arch=('x86_64')
makedepends=('git')
source=('git://git.0pointer.net/libdaemon.git#commit=9fcc28e0e8f84968d1fb8b6d544a42efb13803ec')
sha512sums=('SKIP')

build() {
	cd "${srcdir}/${pkgname}"
	./bootstrap.sh
	./configure --prefix=/usr --localstatedir=/var --disable-lynx
	make
}

package() {
	cd "${srcdir}/${pkgname}"
	make DESTDIR="${pkgdir}" install
}

